Time-delay Example 3

Example from the paper by C. Califano, S. Li, C.H. Moog.

Let us consider the following differential time-delay nonlinear system:

```
eqs = {
    x1'[t] -> x2[t - 1] u[t],
    x2'[t] → x3[t] u[t],
    x3'[t] → u[t]};
vars = {x1[t], x2[t], x3[t], u[t]};
TableForm[eqs]
```

$$\begin{array}{l} x_1'[t] \to u[t]\, x_2[-1+t] \\ x_2'[t] \to u[t]\, x_3[t] \\ x_3'[t] \to u[t] \end{array}$$

Let us introduce the following Ore algebra A of differential time-delay operators

```
replA = ModelToReplacementRules[eqs, t];
A = OreAlgebraWithRelations[Der[t], S[-1][t], replA]
```

$$\mathbb{K}(t)\left[D_t;\ 1,\ D_t\right]\left[\left(S_t^{-1}\right);\ \#1\ /.\ t \to t-1\ \&,\ 0\ \&\right]$$

and the matrix R of ordinary differential time-delay operators which defines the generic linearization of the above nonlinear system

```
MatrixForm[R = ToOrePolynomialD[eqs, vars, A]]
```

$$\begin{pmatrix} D_t & -u[t]\left(S_t^{-1}\right) & 0 & -x_2[-1+t] \\ 0 & D_t & -u[t] & -x_3[t] \\ 0 & 0 & D_t & -1 \end{pmatrix}$$

and the left A-module, finitely presented by R, defined by M = $A^{1\times 4} / A^{1\times 3} R$.

Let us compute the adjoint of R:

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} D_t & 0 & 0 \\ -u[1-t]\left(S_t^{-1}\right) & D_t & 0 \\ 0 & -u[-t] & D_t \\ -x_2[-1-t] & -x_3[-t] & -1 \end{pmatrix}$$

Let us check whether or not M is torsion-free:

```
MatrixForm /@ ({Ann, Rp, Q} = Exti[Radj, A, 1])
```

A very large output was generated. Showing a sample of it.

$$\left\{\begin{pmatrix} D_t & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & u[t]\, D_t - u'[t] \end{pmatrix},\ \begin{pmatrix} 0 & -1 & x_3[t] & 0 \\ 0 & 0 & -D_t & 1 \\ D_t & 0 & -u[t]\, x_3[-1+t]\left(S_t^{-1}\right) & -x_2[-1+t] \end{pmatrix},\ \langle\langle 1 \rangle\rangle\right\}$$

show less | show more | show all | set size limit...

```
aut =
  AutonomousElements[R, {dx1[t], dx2[t], dx3[t], du[t]}, τ, A, Relations → True];
```

```
aut[[1]]
```

```
{τ[1][t] → -dx2[t] + dx3[t] x3[t], τ[2][t] → du[t] - dx3'[t],
 τ[3][t] → -du[t] x2[-1 + t] - u[t] dx3[-1 + t] x3[-1 + t] + dx1'[t]}
```

The autonomous elements τ[1], τ[2], τ[3] satisfy the following equations:

```
aut[[2]]
```

```
{τ[1]'[t] == 0, τ[2][t] == 0, -τ[3][t] u'[t] + u[t] τ[3]'[t] == 0}
```

The A-linear relations among the autonomous elements are given by:

```
aut[[3]]
```

```
{u[t] τ[1][-1 + t] + τ[3][t], -x3[t] τ[2][t] - τ[1]'[t], -τ[2][t]}
```

Let us now prove that the set of autonomous elements can be generated by τ[1] (our guess or inspection of aut[[3]]). Let us introduce the matrix L, defining aut[[3]]:

```
MatrixForm[L = ToOrePolynomialD[aut[[3]], {τ[1][t], τ[2][t], τ[3][t]}, A]]
```

$$\begin{pmatrix} u[t]\,(S_t^{-1}) & 0 & 1 \\ -D_t & -x_3[t] & 0 \\ 0 & -1 & 0 \end{pmatrix}$$

Let us consider the following row vector

```
γ = {{1, 0, 0}};
```

which corresponds to the position of τ[1]. To express τ[2] and τ[3] in terms of τ[1] we first check whether or not the matrix Q formed by stacking L with γ admits a left inverse.

```
S1 = LeftInverse[Q = Join[L, γ], A]
```

$\{\{0, 0, 0, 1\}, \{0, 0, -1, 0\}, \{1, 0, 0, -u[t]\,(S_t^{-1})\}\}$

If we consider the last column of the left inverse S1 of Q, i.e.

```
S2 = List /@ Transpose[S1][[-1]]
```

$\{\{1\}, \{0\}, \{-u[t]\,(S_t^{-1})\}\}$

then we obtain:

```
Thread[{τ[1][t], τ[2][t], τ[3][t]} -> ApplyMatrix[S2, {τ[1][t]}]]
```

```
{τ[1][t] → τ[1][t], τ[2][t] → 0, τ[3][t] → -u[t] τ[1][-1 + t]}
```

From this point we will use some procedures which are not freely available (see NLControl website http://www.nlcontrol.ioc.ee).
Let us integrate the one-form defined by τ[1], namely

```
difs = {ApplyMatrixD[Rp[[1]], vars]}
```

```
{De[-1, {x2[t]}] + De[x3[t], {x3[t]}]}
```

```
BookForm[sp = SpanK[difs, t]]
```

```
SpanK[-dx2[t] + x3[t] dx3[t]]
```

```
Integrability[sp]
```

True

```
IntegrateOneForms[sp]
```

$\left\{x_2[t] - \frac{1}{2} x_3[t]^2\right\}$

We finally obtain the above autonomous element of the nonlinear system.